

**Computation of the
1982 Kimberly-Penman
and the
Jensen-Haise
Evapotranspiration Equations
as Applied in the
U.S. Bureau of Reclamation's
Pacific Northwest AgriMet Program**

February 1994

Prepared by

**Doug Dockter
U. S. Bureau of Reclamation
Pacific Northwest Region
Water Conservation Center**

**Revised January, 2008
Peter L. Palmer
AgriMet Program Manager**

RECLAMATION
Managing Water in the West

Table of Contents

	Page
INTRODUCTION	1
THE 1982 KIMBERLY-PENMAN EQUATION	2
Net Radiation (<i>the heat function</i>).....	3
The slope of the saturation vapor pressure-temperature curve,	3
The psychrometric constant	4
The net radiation energy	5
The soil heat flux	8
Advective Energy Transfer (<i>the wind function</i>)	8
The constant of proportionality.....	9
The dimensionless wind function	9
The mean daily vapor pressure deficit.....	10
The Conversion of the Kimberly-Penman ET_r in $MJ/m^2/d$ to inches/d.....	10
THE JENSEN-HAISE EQUATION.....	11
The temperature coefficient	11
Temperature, $(T_{mean} - T_x)$	12
The global solar radiation, R_s	13
The Conversion of Jensen-Haise ET_r in $cal/cm^2/d$ to inches/d	13
APPENDIX A - Subroutine ETPCOMP.....	14
APPENDIX B - Program JENSEN.....	22
APPENDIX C - Data File CLIMAT.DAT	24
REFERENCES	26

INTRODUCTION

AgriMet is a cooperative agricultural weather station network, managed and operated by the U.S. Bureau of Reclamation, dedicated to agricultural crop water use modeling. The program supports improved irrigation management and scheduling using the Kimberly-Penman combination equation to model and calculate daily water use for a variety of agricultural crops grown in the Pacific Northwest. The Kimberly-Penman equation was determined to be applicable to the area served by AgriMet based on research performed by Reclamation and the Agricultural Research Service (ARS) in Kimberly, Idaho during the early 1970's. This report documents the computation of the 1982 Kimberly-Penman and the Jensen-Haise equations as they are applied in the U.S. Bureau of Reclamation's Pacific Northwest AgriMet Program. (Note: the Jensen-Haise procedure was discontinued in 1994).

AgriMet uses the 1982 Kimberly-Penman equation for computing reference ET, adapted by Dr. James L. Wright of the USDA Agricultural Research Service through his research performed in Kimberly, Idaho (Jensen et al., 1990). This procedure requires several meteorological inputs for modeling ET, including maximum and minimum daily air temperatures, relative humidity, daily solar radiation, and daily wind run. All of these parameters are collected by the AgriMet system. ETr calculated by AgriMet is based on a daily (not hourly) time step. The 1982 Kimberly-Penman model uses alfalfa as the reference crop with reference conditions defined as a well-watered alfalfa crop with 30 to 50 cm of top growth.

Because of the variability of ET rates from crop to crop and the complexity of modeling ET, the accepted standard for deriving crop specific ET (ETc) is to model ET for a reference crop, such as alfalfa (ETr), and then apply this reference ETr value to specific crops through the use of crop coefficients (Kc). These crop coefficients are unique to the reference crop and the individual specific crop, and they vary through time with the growth stage of the plant. These crop coefficients typically are expressed as percent of water use compared to the reference crop. Most of the crop coefficients used by AgriMet were developed through field research conducted by the USDA Agricultural Research Service at Kimberly, Idaho.

The Kimberly-Penman and Jensen-Haise computations are performed in the FORTRAN subroutine program ETPCOMP.FOR (ETPCOMP) (see Appendix A). The weather data required for the computations are supplied via satellite from AgriMet weather stations located throughout the Pacific Northwest. The AgriMet weather data inputs for temperature, wind run, and solar radiation are converted from English to metric SI units in ETPCOMP to accommodate the use of the Kimberly-Penman equation with a minimum of conversion constants.

Daily Jensen-Haise values are calculated and reported within the AgriMet database prior to 1995 as a reference for users and are not used in the AgriMet crop models.

THE 1982 KIMBERLY-PENMAN EQUATION

Since it was first derived by H. L. Penman in 1948, the original Penman equation has been modified a number of times. The AgriMet program uses the 1982 Kimberly-Penman equation adapted by Dr. James L. Wright of the ARS through his research performed in Kimberly, Idaho. The 1982 Kimberly-Penman uses alfalfa as the reference crop with reference conditions established as well-watered with 30 to 50 cm of top growth. The Penman equation used under AgriMet program crop modeling is as follows:

$$\lambda ET_r = \frac{\Delta}{\Delta + \gamma} (R_n - G) + \frac{\gamma}{\Delta + \gamma} 6.43 W_f (e_s - e_a)$$

with units in MJ/m²/d. ETPCOMP will not compute values less than zero.

(Note: (R_n-G) differs from the (R_n+G) term seen in some publications, due to the fact that G is calculated here with an opposite sign convention.)

The expression for the Penman equation appears in ETPCOMP under the variable name ETP_LAMB.

The Penman equation is known as a "combination equation" because it combines **net radiation** (*the heat function*) and **advective energy transfer** (*the wind function*) into one energy balance equation.

The sum of the two terms

$$\frac{\Delta}{\Delta + \gamma} \text{ and } \frac{\gamma}{\Delta + \gamma}$$

is equal to one. These terms represent weighting factors that assess the relative effects of the heat function and the wind function on evaporation. These weightings are approximately 75% *heat function* and 25% *wind function* for the general climatic conditions during the growing season in the Pacific Northwest.

The weighting factors for the *heat function* and the *wind function* appear in ETPCOMP under the variable names FACT1 and FACT2 respectively.

Net Radiation (*the heat function*)

The *heat function* portion of the equation consists of:

$$\frac{\Delta}{\Delta + \gamma} (R_n - G)$$

where:

Δ is the slope of the saturation vapor pressure-temperature curve

γ is the psychrometric constant

R_n is the net radiation

G is the soil heat flux

The slope of the saturation vapor pressure-temperature curve, Δ , represents the change in saturation vapor pressure with respect to temperature (kPa/°C). It is derived by differentiating the following expression, presented by J. F. Bosen in 1960, that estimates the saturation vapor pressure over water (kPa) (for -51 °C < T < 54 °C):

$$e^s = 3.38639[(0.00738T + 0.8072)^8 - 0.000019/1.8T + 48/+0.001316]$$

By making the substitution, $U = 0.00738T + 0.8072$, and taking the derivative of the above equation with respect to T,

$$\Delta = 3.38639(0.05904(0.00738T + 0.8072)^7 - 0.0000342)$$

Further simplifying this, and substituting the mean daily temperature for the temperature variable T, the equation becomes

$$\Delta = 0.200(0.00738T_{mean} + 0.8072)^7 - 0.000116$$

(Note: this equation is valid for $T \geq -23$ °C, however increasing errors occur for temperatures below about -6 °C, due to changes in the relationship between vapor pressure and temperature under these conditions.)

The expression for Δ appears in ETPCOMP under the variable name DD and uses the mean daily temperature in °C (converted from AgriMet data in °F).

The psychrometric constant, γ , (kPa/°C) represents a balance between the sensible heat gained from air flowing past a wet bulb thermometer and the sensible heat transformed into latent heat (Brunt, 1952). It varies directly with air pressure and is calculated as follows:

$$\gamma = \frac{c_p P}{0.622 \lambda}$$

where:

c_p is the specific heat of air at constant pressure and is equal to 0.001005 MJ/kg/°C.

P is the estimated mean atmospheric pressure in kPa.

0.622 is the ratio of the molecular weights of water vapor to air.

λ is the latent heat of vaporization of water in MJ/kg calculated at the mean daily temperature.

P , the mean atmospheric pressure in kPa, is found by applying an equation derived from the ideal gas law that is used for estimating pressure variation with altitude in the Troposphere:

$$P = P_o \left[\frac{T_o - \alpha E}{T_o} \right]^{g/(\alpha R)}$$

where:

P_o is the standard atmospheric pressure and is equal to 101.3 kPa.

T_o is the absolute temperature of a standard atmosphere and is equal to 288 °K.

E is the elevation of the weather station in meters above mean sea level and is stored under the variable name ELEV in the data file CLIMAT.DAT (see Appendix C).

α is the lapse rate, the rate at which the temperature decreases linearly with increasing elevation and is equal to 0.0065 °K/m.

g is the acceleration of gravity and is equal to 9.8067 m/sec².

R is the gas constant and is equal to 287 J/kg/°K for dry air.

Substituting these values into the pressure equation above, the expression becomes:

$$P = 101.3 \left(\frac{288 - 0.0065E}{288} \right)^{5.257}$$

λ , the latent heat of vaporization of water in MJ/kg, is calculated at the mean daily temperature in °C (converted from AgriMet data in °F) as follows:

$$\lambda = 2.501 - 0.002361T_{mean}$$

The expressions for γ , P, and λ appear in ETPCOMP under the variable names GG, PP, and PLAMBDA respectively.

The net radiation energy, R_n , (MJ/m²/d) received by a well-watered irrigated crop has been shown to have a close relationship to the ET rate, and is calculated as follows:

$$R_n = (1 - \alpha)R_s - R_b$$

where:

α is the reflected shortwave (solar) radiation, albedo, and is calculated as:

$$\alpha = 0.29 + 0.06\sin(x + 97.92)$$

(Note: The expression relates the angle of the sun to the time of year. This expression differs slightly from other published versions to accommodate the use of the day number (x) in the year (Dr. James L. Wright, personal contact February, 1994). (1- α) represents the percentage of solar radiation received by the crop-soil surface).

R_s is the measured solar (shortwave) radiation value in MJ/m²/d (converted from AgriMet weather station pyranometer data in cal/cm²/d).

R_b is the estimated outgoing thermal (longwave) radiation in MJ/m²/d and is calculated as follows:

$$R_b = R_{bo} \left(a \frac{R_s}{R_{so}} + b \right)$$

where:

The term $a(R_s/R_{so})+b$ is a factor that adjusts the net clear day outgoing longwave radiation for cloud conditions.

a and b are empirical constants determined from research at Kimberly, Idaho.

$$a = 1.126 \text{ and } b = -0.07 \text{ for } R_s/R_{so} > 0.70$$

$$a = 1.017 \text{ and } b = -0.06 \text{ for } R_s/R_{so} \leq 0.70$$

R_s is the measured shortwave radiation value in MJ/m²/d (converted from AgriMet weather station pyranometer data in cal/cm²/d).

R_{so} is the measured clear-day shortwave radiation in cal/cm²/d. It is represented by a polynomial equation

$$R_{so} = C_1 + C_2x + C_3x^2 + C_4x^3 + C_5x^4$$

where:

x is the day number in the year and the coefficients are unique for each AgriMet weather station site and are read in from CLIMAT.DAT (see Appendix C). This calculation is performed in English units and then converted to MJ/m²/d since the coefficients at each AgriMet site were originally determined in English units.

R_{bo} is the theoretical net outgoing longwave radiation expected to occur on a clear day in

$$R_{bo} = (a_1 + b_1 \sqrt{e_a}) 4.903x 10^{-9} \left(\frac{T_{\max}^4 + T_{\min}^4}{2} \right)$$

MJ/m²/d and is calculated as follows:

where:

The term $a_1 + b_1 \sqrt{e_a}$ is the net emittance expression and represents the difference between the longwave emittance for the reference crop and the effective emittance for the atmosphere.

a_1 and b_2 are empirical constants determined from research at Kimberly, Idaho. They account for the fact that water vapor, smoke and dust particles absorb shortwave radiation from the sun and emit it as longwave

radiation according to their characteristic temperatures.

a_1 is calculated as follows:

$$a_1 = 0.26 + 0.1 e^{-(.0154(x-177))^2}$$

(Note: This expression differs slightly from other published versions to accommodate the use of the day number (x) in the year (Dr. James L. Wright, personal contact February, 1994).

$$b_1 = -0.139$$

e_a is the saturation vapor pressure (kPa) obtained at the mean daily dewpoint temperature using the Bosen equation,

$$e_a = 3.38639[(0.00738T_{\text{dew}} + 0.8072)^8 - 0.000019|1.8T_{\text{dew}} + 48| + 0.001316]$$

T_{dew} is the mean daily dewpoint in °C (converted from the AgriMet data in °F).

4.903×10^{-9} in MJ/m²/d/°K⁴ is the Stefan-Boltzman constant. It is an empirical coefficient representing the proportionality of the emissivity of a body to temperature.

T_{max} and T_{min} are the daily maximum and minimum temperatures in °K (converted from the AgriMet data in °F).

The expressions for R_n , R_s , R_b and R_{bo} appear in ETPCOMP under the variable names RN, SOLAR, RB, and RBO respectively.

The expression for R_{so} appears in ETPCOMP under the variable name CLRAD, with the value converted to MJ/m²/d and stored under the variable name RSO.

The term $a_1 + b_1 \sqrt{e_a}$ appears in ETPCOMP under the variable name EMIT.

The expression for e_a appears in ETPCOMP under the variable name VAPDP.

The soil heat flux, G, (MJ/m²/d) is determined from an empirical equation developed through research conducted on soils at Kimberly, Idaho. It is based on an assumed 50% available moisture content in the soil. This translates to about 24% soil water content by volume or 16% soil water content by weight for a silt-loam soil. The soil heat flux is calculated as follows:

$$G = 0.377(T_{\text{mean}} - T_{\text{pr}})$$

where:

T_{mean} is the mean daily temperature in °C (converted from AgriMet data in °F).

T_{pr} is the average of the previous three days mean temperatures °C (converted from AgriMet data in °F).

(Note: This expression differs slightly from other published versions to accommodate the use of metric units (Dr. James L. Wright, personal contact February, 1994). Under this sign convention, G is positive when the soil is warming. G is subtracted from R_n in the heat function portion of the Penman equation since, if the soil is warming, then part of the net radiation is being used to warm the soil and less is available to evaporate water).

The expressions for G, T_{mean} and T_{pr} appear in ETPCOMP under the variable names HTFLUX, TMEAN, and TPRAV respectively.

Advective Energy Transfer (*the wind function*)

The *wind function* portion of the equation consists of:

$$\frac{\gamma}{\Delta + \gamma} 6.43 W_f (e_s - e_a)$$

where:

$\gamma/(\gamma+\Delta)$ is the wind function weighting factor made up of γ , the psychrometric constant and Δ , the slope of the saturation vapor pressure-temperature curve as described above.

6.43 is the constant of proportionality in MJ/m²/d/kPa.

W_f is the dimensionless wind function.

$(e_s - e_a)$ is the mean daily vapor pressure deficit in kPa.

The constant of proportionality, 6.43 in MJ/m²/d/kPa, relates the dimensionless wind function to the vapor pressure deficit in the equation.

The dimensionless wind function characterizes the effects of wind in the advection (motion of the atmosphere) of sensible heat for ET and is calculated as follows:

$$W_f = a_w + b_w U_2$$

where:

U_2 is the wind speed in kilometers per day (converted from AgriMet data in miles per day).

a_w and b_w are empirical coefficients that lessen the effect of the wind function during the early part of the year. These are based on the assumption that the surrounding rangeland is wetter early in the season and has less advection of sensible heat than occurs later in the year when surrounding areas are drier. These coefficients assume an anemometer height of 2 meters.

a_w and b_w are calculated as follows with x representing the day number in the calendar year:

$$a_w = 0.4 + 1.4 e^{-\left(\frac{x-173}{58}\right)^2}$$

$$b_w = 0.007 + 0.004 e^{-\left(\frac{x-243}{80}\right)^2}$$

The expressions for W_f , a_w , b_w , and U_2 appear in ETPCOMP under the variable names WF, AW, BW, and U2 respectively.

The mean daily vapor pressure deficit is expressed by the quantity ($e_s - e_a$) in kPa.

where:

The term e_s is the average saturation vapor pressure (kPa) calculated from the saturation vapor pressures determined at the maximum and minimum daily temperatures as follows:

$$e_s = \frac{e_{T_{max}} + e_{T_{min}}}{2}$$

where:

$e_{T_{max}}$ and $e_{T_{min}}$ are in kPa and are determined from the Bosen equation at the maximum and minimum daily air temperatures, T_{max} and T_{min} in °C (converted from AgriMet data in °F) as follows:

$$e_{T_{max}} = 3.38639[(0.00738T_{max} + 0.8072)^8 - 0.000019|1.8T_{max} + 48| + 0.001316]$$

$$e_{T_{min}} = 3.38639[(0.00738T_{min} + 0.8072)^8 - 0.000019|1.8T_{min} + 48| + 0.001316]$$

e_a is the saturation vapor pressure (kPa) determined from the Bosen equation at the mean daily dewpoint temperature, T_{dew} , in °C (converted from AgriMet data in °F) as follows:

$$e_a = 3.38639[(0.00738T_{dew} + 0.8072)^8 - 0.000019|1.8T_{dew} + 48| + 0.001316]$$

The expressions for $e_{T_{max}}$, $e_{T_{min}}$, and e_a appear in ETPCOMP under the variable names VAPMX, VAPMN, and VAPDP respectively.

The Conversion of the Kimberly-Penman ET_r in $MJ/m^2/d$ to inches/d

Conversion of the units to inches per day is accomplished by multiplying the results of the equation in $MJ/m^2/d$ by the following:

$$\frac{(1 \times 10^3 \frac{\text{cm}^3}{\text{kg}})(1 \times 10^{-6} \frac{\text{m}^3}{\text{cm}^3})(39.37 \frac{\text{inches}}{\text{m}})}{\lambda \frac{\text{MJ}}{\text{kg}}}$$

where the latent heat of water in MJ/kg is $\lambda = 2.501 - 0.002361T_{\text{mean}}$.

THE JENSEN-HAISE EQUATION

The Jensen-Haise equation (cal/cm²/d) represents a temperature-radiation method of calculating a daily ET_r. Although the AgriMet program uses the 1982 Kimberly-Penman for crop modeling, it also calculates Jensen-Haise values as an additional reference (also using alfalfa as the reference crop with conditions established as well watered with 30 to 50 cm of top growth). English units are used in the calculation since the coefficients have been determined for each individual weather station site in English units. The Jensen-Haise equation is as follows:

$$\lambda ET_r = C_T (T_{\text{mean}} - T_x) R_s$$

where:

C_T is the temperature coefficient.

T_{mean} is the daily mean temperature in °F.

T_x is the intercept of the temperature axis as calculated for each AgriMet weather station.

R_s is the measured global solar radiation in cal/cm²/day

The product of $C_T(T_{\text{mean}} - T_x)$ represents a weighting function on R_s .

The expression for the Jensen-Haise equation appears in ETPCOMP under the variable name ETPJ_VALUE.

The temperature coefficient, C_T , is an empirical coefficient based on temperature and atmospheric pressure. It relates temperature and radiation to ET. C_T is unique for each AgriMet weather station site and is calculated using the FORTRAN program JENSEN.FOR (JENSEN)(see Appendix B). The values for each station are stored under the variable name CSUBT in a file that is accessed by ETPCOMP called CLIMAT.DAT (see Appendix C). C_T is calculated as follows:

$$C_T = \frac{I}{C_1 + C_2 C_H}$$

where:

C_1 , C_2 , and C_H are empirically derived correlation coefficients as follows:

where:

ELEV is the elevation of the AgriMet weather station in feet above mean sea level (from prompted input to JENSEN).

$$C_1 = 68 - 3.6 \frac{ELEV}{1000}$$

$$C_2 = 13^\circ\text{F}$$

$$C_H = \frac{50 \text{ milibars}}{e_2 - e_1}$$

where:

e_2 and e_1 are vapor pressures determined at T_{\max} and T_{\min} and are calculated as follows (e_1 is the same equation except for the substitution of T_{\min} for T_{\max}):

$$e_2 = 6.105 e^{\left(25.22 \frac{\left(\frac{5}{9}(T_{\max}-32)+273\right)-273}{\frac{5}{9}(T_{\max}-32)+273} - 5.31 \ln\left(\frac{\frac{5}{9}(T_{\max}-32)+273}{273}\right)\right)}$$

where:

T_{\max} and T_{\min} are the long term mean maximum and mean minimum air temperatures respectively for the month of the year with the highest mean air temperatures in °F (from prompted input to JENSEN).

Temperature, ($T_{\text{mean}} - T_x$), is the difference between T_{mean} , the daily mean temperature (from AgriMet data in °F), and T_x , the intercept of the temperature axis. T_x is unique for each AgriMet weather station site and is calculated in JENSEN. The values for each station are stored under the variable name TSUBX in CLIMAT.DAT. T_x is calculated as follows:

$$T_x = 27.5 - 0.25(e_2 - e_1) - \frac{ELEV}{1000}$$

where:

ELEV is the elevation of the AgriMet weather station in feet above mean sea level (from prompted input to JENSEN).

The global solar radiation, R_s , is the measured solar radiation value in cal/cm²/d from AgriMet weather station pyranometer data. It appears in ETPCOMP under the variable name SR_VALUE.

The Conversion of Jensen-Haise ET_r in cal/cm²/d to inches/d

Conversion of the units to inches/d is accomplished by multiplying the results of the equation in cal/cm²/d by

$$\frac{0.3937 \frac{\text{inches}}{\text{cm}}}{\left(\lambda \frac{\text{cal}}{\text{g}}\right) \left(1 - \frac{\text{g}}{\text{cm}^3} \text{ of water}\right)}$$

where:

The latent heat of water in cal/g is $\lambda = 595 - 0.51T_{\text{mean}}$. T_{mean} is in °C (converted from AgriMet data in °F) for this calculation.

APPENDIX A - Subroutine ETPCOMP

```
C
C
C   XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX
C   XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX
C
C   ROUTINE TITLE:  SUBROUTINE ETPCOMP (FEB 1994)
C   PREVIOUS VERSION SAVED AS ETPCOMP.OLD
C
C   PURPOSE:  COMPUTES EVAPOTRANSPIRATION RATE USING THE 1982
C             KIMBERLY-PENMAN EQUATION AND THE JENSEN-HAISE
C             EQUATION.
C
C   AUTHOR:  JOHN HOCKETT - USBR
C
C           MODIFICATION - KEEFER - NOVEMBER 87
C
C             FOUND THE FILE CLIMAT.DAT AND MOVED IT TO
C             THE DMS SYSTEM FROM OLD BMHS.  MODIFIED THE
C             LIST-DIRECTED READ FOR THE 5 CONSTANTS SO
C             THE FILE WOULD OPEN AND READ CORRECTLY.
C
C           KEEFER - MAR 88
C
C             ADDED LIMITS CHECKS FOR ALL INPUT VARIABLES
C
C           POWERS - JAN 89
C
C             MODIFIED GAMM1 AND GAMM2 CALCULATION PROCEDURE
C             BASED ON UPDATED ELEVATION FUNCTION AS PER
C             JIM WRIGHT (IN ADDITION, THIS VERSION SET UP
C             WITHIN DUA12:[APOWERS.ET] INDEPENDENT OF DMS TO
C             BE AUTOMATICALLY RUN USING ARCHIVE DATA THROUGH
C             DUA12:[APOWERS.ET]AUTOACMET.COM
C             GAMM1 AND GAMM2 ARE NOW CALLED FACT1 AND FACT2
C             (DOCKTER - FEB 1994)
C
C           BROWER - APR 89
C
C             MODIFIED TO CALCULATE THE JENSEN ETP VALUE,
C             CALLED ETPJ_VALUE.
C
C           POWERS - APR 89
C
C             PROGRAMS ETPCOMP,ACMET,ACMETALL,AUTOACMET &
C             CLIMAT.DAT MODIFIED AND MOVED TO OPERATE
C             FROM DUA12:[AGRIMET.ET]
C
C           POWERS - MAY 89
```



```

C
C   MM_SAVE          = PREVIOUS DAYS MEAN TEMP IN F
C   OB_MEAN         = TODAY MEAN TEMP IN F
C   WR_VALUE        = WIND RUN
C   OB_MAX          = TODAYS MAX TEMP IN F
C   OB_MIN          = TODAYS MIN TEMP IN F
C   TP_VALUE        = TODAYS DEW POINT IN F
C   SR_VALUE        = SOLAR RADIATION
C   NOVAL           = 998877.00
C   MONTH           = MONTH NAME
C   DAY             = CHARACTER DAY NUMBER
C   STATION         = COLUMBIA BASIN TELETYPE (CBTT) STATION NAME
C   LOGUNIT         = OUTPUT UNIT FOR ERROR MESSAGES
C
C   REAL KMIN,KMAX,JLAMBDA
C   REAL*4 C(4),MM_SAVE(3),CLDYC(5),NOVAL
C   INTEGER*2 KOUNT
C   INTEGER*4 IEND(13)
C   CHARACTER MONTH*3,DAY*2,MONTHS_ALL*48,STATION*12,STN*12
C
C   DATA MONTHS_ALL/'JAN*FEB*MAR*APR*MAY*JUN*JUL*AUG*
1   SEP*OCT*NOV*DEC*'/
C   DATA IEND/0,31,59,90,120,151,181,212,243,273,304,
1   334,365/
C
C   XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX
C   BEGIN EXECUTABLE CODE
C   XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX
C
C   NOVAL=998877
C   ETP_VALUE=NOVAL
C   ETPJ_VALUE=NOVAL
C   IF(OB_MEAN.EQ.NOVAL.OR.TP_VALUE.EQ.NOVAL.OR.
1   WR_VALUE.EQ.NOVAL.OR.SR_VALUE.EQ.NOVAL.OR.OB_MIN.
2   EQ.NOVAL.OR.OB_MAX.EQ.NOVAL)THEN
C   RETURN          ! NOT ENOUGH DATA TO CALC ETP
C   END IF
C
C   LIMIT WIND RUN TO 150 MILES PER DAY.
C
C   IF(WR_VALUE.GT.150.0)WR_VALUE=150.0
C
C   READ CONSTANTS FROM CLIMAT.DAT --
C   STN = WEATHER STATION COLUMBIA BASIN TELETYPE CODE
C   CLDYC= CONSISTS OF 5 CLEARDAY SOLAR RADIATION COEFFICIENTS
C   CLDYM= THE CLEAR DAY MINIMUM SOLAR RADIATION
C   ANEMH= ANEMOMETER HEIGHT - THIS VALUE IS NO LONGER USED IN
C   THE COMPUTATIONS SINCE ALL ANEMOMETERS IN THE FIELD
C   HAVE BEEN SET TO APROX. 2 METERS IN HEIGHT AS IS
C   ASSUMED BY THE 1982 KIMBERLY-PENMAN EQU. CLIMAT.DAT

```

```

C          SHOWS THESE STATIONS TO BE AT 7FT WHICH IS OUTDATED
C          INFO AT PRESENT
C      ELEV = WEATHER STATION ELEVATION IN METERS
C          CSUBT= TEMPERATURE COEFFICIENT JENSEN-HAISE EQU.
C          TSUBX= INTERCEPT OF THE TEMPERATURE AXIS FOR JENSEN-HAISE
C          EQUATION
C
C          OPEN(UNIT=76,FILE='DUA12:[AGRIMET.ET]CLIMAT.DAT',TYPE='OLD',
1          READONLY,ERR=1000)
30      READ(76,'(A4,3F9.0,F10.0,F9.0,F6.0,F4.0,2F7.0,F8.0)',ERR=1000,
2      END=1000,
3          IOSTAT=ISTAT)STN,CLDYC,CLDYM,ANEMH,ELEV,CSUBT,TSUBX
      IF(STN.NE.STATION)GO TO 30
C
C          CONVERT WEATHER DATA TO SI UNITS
C
C          TMEAN=5.*(OB_MEAN-32.)/9.
C          TMAX=5.*(OB_MAX-32.)/9.
C          TMIN=5.*(OB_MIN-32.)/9.
C          TDEW=5.*(TP_VALUE-32.)/9.
C          KMAX=TMAX+273.16
C          KMIN=TMIN+273.16
C          SOLAR=SR_VALUE*0.041868
C          U2=WR_VALUE*1.6093
C
C          COMPUTE LAMBDA IN CAL/G FOR JENSEN
C
C          JLAMBDA=595.-0.51*TMEAN
C
C          COMPUTE JENSEN ETP - THIS IS IN ENGLISH UNITS BECAUSE
C          CSUBT AND TSUBX FOR ALL STATIONS (CLIMAT.DAT) HAVE BEEN
C          CALCULATED IN ENGLISH UNITS.  CHANGED FORMULA TO USE
C          OB_MEAN INSTEAD OF CALCULATING AN AVERAGE TEMPERATURE
C          (FEB 94-DDOCKTER)
C
C          ETPJ_VALUE=CSUBT*(OB_MEAN-TSUBX)*SR_VALUE*0.3937/JLAMBDA
C
C          BEGIN COMPUTATIONS FOR PENMAN ET
C
C          COMPUTE LAMBDA FOR PENMAN IN MJ/kg, DELTA (DD) IN kPa/deg C,
C          ATMOS. PRESSURE (PP) IN kPa, GAMMA (GG) IN kPa/deg C,
C          HEAT AND WIND FUNCTION WEIGHTING FACTORS (FACT1 AND FACT2)
C          ARE DIMENSIONLESS.
C
C          PLAMBDA=2.501-0.002361*TMEAN
C          DD=0.2*(0.00738*TMEAN+0.8072)**7.-0.000116
C          PP=101.3*((288.-0.0065*ELEV)/288.)**5.257)
C          GG=0.001005*(PP)/(0.622*PLAMBDA)
C          FACT1=DD/(DD+GG)
C          FACT2=1.-FACT1

```

```

C
C   COMPUTE AVERAGE TEMP FOR PRIOR THREE DAYS USING MM
C   IF PRIOR DAYS ARE NOVAL THEN USE TODAY'S MM
C
PREAV=0.0
KOUNT=0
DO 10 L=1,3
IF(MM_SAVE(L).EQ.NOVAL)GO TO 10
PREAV=PREAV+MM_SAVE(L)
KOUNT=KOUNT+1
10 CONTINUE
IF(KOUNT.EQ.0)THEN
PREAV=OB_MEAN
ELSE
PREAV=PREAV/FLOATI(KOUNT)
END IF

C
C   CONVERT PREAV TO CELSIUS
C
TPREAV=5.*(PREAV-32.)/9.

C
C   FIND HEAT FLUX, HTFLUX
C
HTFLUX=(TMEAN-TPREAV)*0.377

C
C   FIND MIN VAPOR PRESSURE
C
VAPMN=3.38639*((.00738*TMIN+.8072)**8.-.000019*ABS(1.8*TMIN+48.)
1 + 0.001316)

C
C   FIND MAX VAPOR PRESSURE
C
VAPMX=3.38639*((.00738*TMAX+.8072)**8.-.000019*ABS(1.8*TMAX+48.)
1 + 0.001316)

C
C   FIND DEW POINT VAPOR PRESSURE
C
VAPDP=3.38639*((.00738*TDEW+.8072)**8.-.000019*ABS(1.8*TDEW+48.)
1 + 0.001316)

C
C   FIGURE OUT THE DAY NUMBER IN THE CALENDAR YEAR
C
MO=INDEX(MONTHS_ALL,MONTH)/4+1
READ(UNIT=DAY,FMT=20,ERR=1000)IDAY
20 FORMAT(I2)
MDAY=IEND(MO)+IDAY
X=MDAY

C
C   FIND CLEAR DAY SOLAR RADIATION IN CAL/SQ CM/DAY (CLRAD)
C   AND CONVERT TO MJ/SQ METER/DAY (RSO). THIS IS IN

```

```

C          CAL/SQ CM/DAY BECAUSE THAT IS HOW COEFFICIENTS WERE
C          ORIGINALLY DETERMINED FOR EACH STATION IN POLYNOMIAL
C          EQUATION FROM CLIMAT.DAT.
C
CLRAD=CLDYC(1)+CLDYC(2)*X+CLDYC(3)*X**2.+
1      CLDYC(4)*X**3.+CLDYC(5)*X**4.
IF (CLRAD.LT.CLDYM)CLRAD=CLDYM
      RSO=CLRAD*.041868
C
C      SOLVE FOR NET EMITTANCE
C
      A1=.26+.1*EXP(-1.*( .0154*(X-177.))**2.)
      B1=-0.139
      EMIT=A1+B1*SQRT(ABS(VAPDP))
C
C      SOLVE FOR NET OUTGOING LONG WAVE RADIATION MJ/SQ M/DAY
C
RBO=EMIT*(4.903E-9)*(KMAX**4.+KMIN**4.)/2.
C
C      SOLVE FOR SHORTWAVE (SOLAR) RADIATION REFLECTANCE (ALBEDO)
C
      ALBEDO=.29+.06*SIND(X+97.92)
C
C      SOLVE FOR CLOUD COVER EXPRESSION
C
IF(RSO.EQ.0.0)RSO=.000001
SOL=SOLAR/RSO
IF(SOL.GT.1.0)SOL=1.0
  IF(SOL.GT.0.70)THEN
    A=1.126
    B=-0.07
  ELSE
    A=1.017
    B=-0.06
  END IF
C
C      COMPUTE ESTIMATED LONGWAVE (THERMAL) OUTGOING RADIATION
C
RB=RBO*(A*SOL+B)
C
C      COMPUTE NET INCOMING RADIATION
C
RN=(1.-ALBEDO)*SOLAR-RB
C
C      COMPUTE VAPOR PRESSURE DEFICIT
C
VAPDE=((VAPMN+VAPMX)*.5)-VAPDP
C
C      COMPUTE WIND FUNCTION (WF) AND COEFFICIENTS (AW) AND (BW)
C      THESE COEFFICIENTS ASSUME ANEMOMETER HT OF 2 METERS (7 FT OK)

```

```

C          IF ANY ARE DIFFERENT IN FUTURE, A HEIGHT COVERSION FUNCTION
C          WILL NEED TO BE ADDED HERE.
C
C          AW=.4+1.4*EXP(-1.*((X-173.)/58.)**2.)
C          BW=.007+.004*EXP(-1.*((X-243.)/80.)**2.)
C          WF= AW+BW*U2
C
C          SOLVE FOR EVAPO RATE PENMAN METHOD
C
C          ETP_LAMB=FACT1*(RN-HTFLUX)+FACT2*6.43*WF*VAPDE
C
C          CONVERT FROM MJ/SQ M/DAY TO INCHES/DAY
C
C          ETP_VALUE=ETP_LAMB*.03937/PLAMBDA
C
C          PRINT VALUES TO SCREEN
C
C          PRINT *, 'OB_MAX', OB_MAX
C          PRINT *, 'TMAX', TMAX
C          PRINT *, 'KMAX', KMAX
C          PRINT *, 'OB_MIN', OB_MIN
C          PRINT *, 'TMIN', TMIN
C          PRINT *, 'KMIN', KMIN
C          PRINT *, 'OB_MEAN', OB_MEAN
C          PRINT *, 'TMEAN', TMEAN
C          PRINT *, 'TP_VALUE', TP_VALUE
C          PRINT *, 'TDEW', TDEW
C          PRINT *, 'MM_SAVE', MM_SAVE
C          PRINT *, 'PREAV', PREAV
C          PRINT *, 'TPREAV', TPREAV
C          PRINT *, 'JLAMBDA', JLAMBDA
C          PRINT *, 'PLAMBDA', PLAMBDA
C          PRINT *, 'DELTA', DD
C          PRINT *, 'GAMMA', GG
C          PRINT *, 'PRESSURE', PP
C          PRINT *, 'FACT1', FACT1
C          PRINT *, 'FACT2', FACT2
C          PRINT *, 'SR_VALUE', SR_VALUE
C          PRINT *, 'SOLAR', SOLAR
C          PRINT *, 'RN', RN
C          PRINT *, 'RB', RB
C          PRINT *, 'RBO', RBO
C          PRINT *, 'CLRAD', CLRAD
C          PRINT *, 'RSO', RSO
C          PRINT *, 'ALBEDO', ALBEDO
C          PRINT *, 'A1', A1
C          PRINT *, 'B1', B1
C          PRINT *, 'A', A
C          PRINT *, 'B', B
C          PRINT *, 'DAY', X

```

```

PRINT *, 'AW', AW
PRINT *, 'BW', BW
PRINT *, 'WF', WF
PRINT *, 'WR_VALUE', WR_VALUE
PRINT *, 'U2', U2
PRINT *, 'HTFLUX', HTFLUX
PRINT *, 'VAPMX', VAPMX
PRINT *, 'VAPMN', VAPMN
PRINT *, 'VAPDP', VAPDP
PRINT *, 'VAPDE', VAPDE
PRINT *, 'EMIT', EMIT
PRINT *, 'ET', ETP_VALUE
PRINT *, 'EJ', ETPJ_VALUE
PRINT *, 'CONSTANTS', STN, CLDYC(1), CLDYC(2), CLDYC(3), CLDYC(4)
1 ,CLDYC(5), CLDYM, ANEMH, ELEV, CSUBT, TSUBX

C
CLOSE(UNIT=76, ERR=1002)

C
IF(ETP_VALUE.LT.0.0)ETP_VALUE=0.0
IF(ETPJ_VALUE.LT.0.0)ETPJ_VALUE=0.0

C
C WRITE ETP AND EJ TO A FILE FOR COMPARISON TO OLD ETP
C
C WRITE(15,9800)ETP_VALUE,ETPJ_VALUE
C9800 FORMAT(1X,2E10.3)
C
RETURN

C
1000 WRITE(6,1001,ERR=925) STATION
1001 FORMAT(' %ETPCOMP-1001-ERROR COMPUTING ET FOR ',A)
925 CLOSE(UNIT=76,ERR=950)
950 RETURN 1

C
1002 RETURN
END

```

APPENDIX B - Program JENSEN

```
C      THIS PROGRAM CALCULATES CSUBT AND TSUBX WHICH ARE USED IN JENSENS
C      EQUATION FOR CALCULATING EVAPOTRANSPIRATION POTENTIAL IN THE
C      IRRIGATION SCHEDULING PROGRAM. INPUT CONSISTS OF LONG TERM MEAN
C      MAXIMUM AND MINIMUM TEMPERATURES IN DEGREE F. FORTTHE MONTH OF
C      HIGHEST MEAN AIR TEMPERATURES, AND THE ELEVATION OF THE WEATHER
C      STATION IN FEET. INPUT IS PROMPTED BY THE PROGRAM.
      PRINT*, 'ENTER MEAN MINIMUM TEMPERATURE'
      READ*, TMIN
      PRINT*, 'ENTER MEAN MAXIMUM TEMPERATURE'
      READ*, TMAX
      PRINT*, 'ENTER ELEVATION IN FEET'
      READ*, ELEV

C
C      OBTAIN SATURATED VAPOR PRESSURE IN MB FOR THE TWO TEMPS
      TEMPF = TMIN
      I=1
100    TEMPC = (5.0 / 9.0) * (TEMPF - 32.0)
      TEMPK = TEMPC + 273.0
      CA = TEMPK / 273.0
      CB = ALOG(CA)
      CC = CB * 5.31
      CD = (TEMPK - 273.0) / TEMPK
      CE = (CD * 25.22) - CC
      VAPORP = 6.105 * EXP(CE)
      IF(I .EQ. 2) GO TO 110
      VP1 = VAPORP
      TEMPF = TMAX
      I=I+1
      GO TO 100
110    VP2 = VAPORP
      VPDIFF = VP2 - VP1

C
C      COMPUTE CSUBH , CSUBT , TSUBX VALUES
C
      CONE = 68.0 - (3.6 * ELEV / 1000.0)
      CSUBH = 50.0 / VPDIFF
      CSUBT = 1.0 / (CONE + 13.0 * CSUBH)
      TSUBX = 27.5 - 0.25 * VPDIFF - ELEV / 1000.0
      PRINT 120, VPDIFF, CONE, CSUBH, CSUBT, TSUBX
      STOP

C
C      FORMATS
C
120    FORMAT(3X, 28HVAPOR PRESSURE DIFFERENCE = ,F12.5/
      +3X, 7HCONE = ,F12.5/ 3X, 8HCSUBH = ,F12.5/
      +3X,
      +50H THE FOLLOWING ARE USED IN THE IRRIGATION PROGRAM, /
```



```
+3X, 14H      CSUBT = ,F12.5/ 3X, 14H      TSUBX = ,F12.5)
END
```

APPENDIX C - Data File CLIMAT.DAT

STN is the weather station code
 CLDYC(1) thru CLDYC(5) are coefficients used in a fifth degree polynomial
 along with the day of the year to compute clear day solar radiation.
 CLDYM is the clear day minimum solar radiation (cal/cm²/d)
 AHT is the anemometer height in feet
 ELEV is the weather station site elevation in meters above mean sea level
 CsubT is used in the Jensen Haise equation
 TsubX is used in the Jensen Haise equation

Data File: CLIMAT.DAT

<u>STN</u>	<u>CLDYC1</u>	<u>CLDYC2</u>	<u>CLDYC3</u>	<u>CLDYC4</u>	<u>CLDYC5</u>	<u>CLDYM</u>	<u>AHT</u>	<u>ELEV</u>	<u>CsubT</u>	<u>TsubX</u>
LIDW	.154E+3	.176E+1	.614E-1	-.396E-3	.590E-6	100.0	7.0	497.0	0.0115	19.3
HERO	.183E+3	.151E+1	.611E-1	-.389E-3	.578E-6	100.0	7.0	199.0	0.0115	19.2
MALI	.213E+3	.960E+0	.599E-1	-.362E-3	.521E-6	100.0	7.0	1345.0	0.0139	15.0
ODSW	.154E+3	.176E+1	.614E-1	-.396E-3	.590E-6	100.0	7.0	503.0	0.0113	19.8
MRSO	.195E+3	.703E+0	.680E-1	-.402E-3	.575E-6	100.0	7.0	744.0	0.0114	19.3
COVM	.165E+3	.187E+1	.584E-1	-.381E-3	.596E-6	100.0	7.0	1096.0	0.0122	17.9
OMAW	.132E+3	.144E+1	.684E-1	-.432E-3	.643E-6	100.0	7.0	376.0	0.0114	19.4
CHVO	.199E+3	.231E+1	.476E-1	-.326E-3	.490E-6	100.0	7.0	1329.0	0.0121	17.8
CEDC	.219E+3	.245E+1	.437E-1	-.305E-3	.460E-6	100.0	7.0	1402.0	0.0124	17.4
BANO	.201E+3	.231E+1	.479E-1	-.328E-3	.493E-6	100.0	7.0	24.0	0.0061	25.7
PMAI	.193E+3	.218E+1	.508E-1	-.342E-3	.514E-6	100.0	7.0	703.0	0.0122	18.0
GERW	.147E+3	.191E+1	.601E-1	-.392E-3	.587E-6	100.0	7.0	351.0	0.0111	20.2
GOLW	.170E+3	.164E+1	.610E-1	-.391E-3	.582E-6	100.0	7.0	512.0	0.0113	19.6
LEGW	.179E+3	.148E+1	.620E-1	-.394E-3	.585E-6	100.0	7.0	259.0	0.0111	20.2
HOXO	.167E+3	.194E+1	.573E-1	-.376E-3	.562E-6	100.0	7.0	171.0	0.0100	22.2
HRHW	.170E+3	.164E+1	.610E-1	-.391E-3	.582E-6	100.0	7.0	274.0	0.0115	19.4
AHTI	.191E+3	.217E+1	.512E-1	-.344E-3	.516E-6	100.0	7.0	1615.0	0.0131	16.2
RXGI	.191E+3	.217E+1	.512E-1	-.344E-3	.516E-6	100.0	7.0	1486.0	0.0131	6.3
FAFI	.199E+3	.231E+1	.476E-1	-.326E-3	.490E-6	100.0	7.0	1536.0	0.0139	15.1
AFTY	.204E+3	.228E+1	.480E-1	-.328E-3	.492E-6	100.0	7.0	1893.0	0.0147	14.1
RPTI	.207E+3	.238E+1	.461E-1	-.319E-3	.480E-6	100.0	7.0	1266.0	0.0137	15.3
ECHO	.183E+3	.151E+1	.611E-1	-.389E-3	.578E-6	100.0	7.0	232.0	0.0116	19.0
HRMO	.183E+3	.151E+1	.611E-1	-.389E-3	.578E-6	100.0	7.0	203.0	0.0116	19.0
LAKO	.219E+3	.245E+1	.437E-1	-.305E-3	.460E-6	100.0	7.0	1436.0	0.0135	15.7
CRSM	.124E+3	.203E+1	.614E-1	-.403E-3	.605E-6	100.0	7.0	899.0	0.0121	18.2
RDBM	.124E+3	.203E+1	.614E-1	-.403E-3	.605E-6	100.0	7.0	927.0	0.0121	18.1
PCYO	.195E+3	.703E+0	.680E-1	-.402E-3	.575E-6	100.0	7.0	1144.0	0.0121	18.0
MDFO	.219E+3	.245E+1	.437E-1	-.305E-3	.460E-6	100.0	7.0	408.0	0.0116	19.0
CRVO	.185E+3	.208E+1	.531E-1	-.354E-3	.530E-6	100.0	7.0	70.0	0.0105	21.4
TWFI	.207E+3	.240E+1	.459E-1	-.318E-3	.478E-6	100.0	7.0	1194.0	0.0129	16.6
ABEI	.204E+3	.228E+1	.480E-1	-.328E-3	.493E-6	100.0	7.0	1332.0	0.0130	16.5
SIGM	.155E+3	.160E+1	.636E-1	-.406E-3	.606E-6	100.0	7.0	895.5	0.0121	18.1
FOGO	.190E+3	.104E+1	.629E-1	-.378E-3	.543E-6	100.0	7.0	54.8	0.0108	21.0
ONTO	.193E+3	.218E+1	.508E-1	-.342E-3	.514E-6	100.0	7.0	688.0	0.0122	18.0

GDVI	.234E+3	.141E+1	.557E-1	-.355E-3	.525E-6	100.0	7.0	786.0	0.0122	18.0
FTHI	.204E+3	.228E+1	.480E-1	-.328E-3	.493E-6	100.0	7.0	1354.0	0.0130	16.5
GFRI	.234E+3	.141E+1	.557E-1	-.355E-3	.525E-6	100.0	7.0	921.4	0.0122	18.0
PICI	.199E+3	.231E+1	.476E-1	-.326E-3	.490E-6	100.0	7.0	1485.0	0.0139	15.1
POBO	.255E+3	.139E+1	.695E-1	-.408E-3	.592E-6	100.0	7.0	974.7	0.0125	17.5
MASW	.141E+3	.156E+1	.655E-1	-.417E-3	.622E-6	100.0	6.6	600.6	0.0120	18.3
IMBO	.180E+3	.169E+1	.602E-1	-.392E-3	.590E-6	100.0	6.6	837.6	0.0124	17.5
BOII	.193E+3	.218E+1	.508E-1	-.342E-3	.514E-6	100.0	6.6	828.5	0.0122	18.0
NMPI	.193E+3	.218E+1	.508E-1	-.342E-3	.514E-6	100.0	6.6	802.8	0.0122	18.0
HRFO	.189E+3	.213E+1	.519E-1	-.348E-3	.522E-6	100.0	6.6	1158.0	0.0124	17.5
DRLM	.161E+3	.175E+1	.606E-1	-.392E-3	.584E-6	100.0	6.6	1418.0	0.0124	17.5
ARAO	.182E+3	.134E+1	.657E-1	-.412E-3	.611E-6	100.0	6.6	42.4	0.0108	21.0
KFLO	.215E+3	.240E+1	.450E-1	-.312E-3	.470E-6	100.0	6.6	1249.7	0.0135	15.7
BRKO	.213E+3	.242E+1	.449E-1	-.312E-3	.470E-6	100.0	6.6	24.0	0.0061	25.7
WRDO	.215E+3	.240E+1	.450E-1	-.312E-3	.470E-6	100.0	6.6	1249.7	0.0135	15.7
AGKO	.215E+3	.240E+1	.450E-1	-.312E-3	.470E-6	100.0	6.6	1249.7	0.0135	15.7
LORO	.215E+3	.240E+1	.450E-1	-.312E-3	.470E-6	100.0	6.6	1268.0	0.0135	15.7
FALN	.246E+3	.269E+1	.369E-1	-.270E-3	.409E-6	100.0	6.6	1208.5	0.0000	00.0
BKVO	.171E+3	.195E+1	.568E-1	-.373E-3	.558E-6	100.0	6.6	1042.4	0.0000	00.0
EURN	.246E+3	.269E+1	.369E-1	-.270E-3	.409E-6	100.0	6.6	1797.4	0.0000	00.0
BNDW	.168E+3	.196E+1	.569E-1	-.374E-3	.559E-6	100.0	6.6	24.1	0.0000	00.0
CJDW	.143E+3	.153E+1	.659E-1	-.418E-3	.623E-6	100.0	6.6	304.0	0.0000	00.0
GCDW	.142E+3	.161E+1	.649E-1	-.414E-3	.617E-6	100.0	6.6	404.0	0.0000	00.0
SBMW	.143E+3	.162E+1	.646E-1	-.413E-3	.616E-6	100.0	6.6	419.0	0.0000	00.0
KFLW	.134E+3	.147E+1	.676E-1	-.427E-3	.636E-6	100.0	6.6	407.0	0.0000	00.0
LBRW	.156E+3	.179E+1	.608E-1	-.394E-3	.588E-6	100.0	6.6	192.0	0.0000	00.0
DENI	.158E+3	.177E+1	.608E-1	-.393E-3	.587E-6	100.0	6.6	506.0	0.0000	00.0
SILW	.160E+3	.180E+1	.600E-1	-.389E-3	.581E-6	100.0	6.6	251.0	0.0000	00.0
DTRO	.178E+3	.210E+1	.538E-1	-.358E-3	.537E-6	100.0	6.6	502.9	0.0000	00.0
HCKO	.192E+3	.222E+1	.507E-1	-.342E-3	.513E-6	100.0	6.6	438.9	0.0000	00.0
LKPO	.192E+3	.222E+1	.507E-1	-.342E-3	.513E-6	100.0	6.6	283.5	0.0000	00.0
BEWO	.188E+3	.215E+1	.519E-1	-.348E-3	.522E-6	100.0	6.6	1104.9	0.0000	00.0
PNGO	.169E+3	.192E+1	.574E-1	-.376E-3	.563E-6	100.0	6.6	189.0	0.0000	00.0
DEFO	.170E+3	.192E+1	.572E-1	-.375E-3	.561E-6	100.0	6.6	384.0	0.0000	00.0
BATO	.206E+3	.241E+1	.464E-1	-.321E-3	.482E-6	100.0	6.6	1317.0	0.0000	00.0
CHAW	.141E+3	.158E+1	.653E-1	-.416E-3	.620E-6	100.0	6.6	594.0	0.0000	00.0

REFERENCES

Am. Soc. Agric. Engrs. (1980). Design and Operation of Farm Irrigation Systems, Am. Soc. Agric. Engrs. Mongraph No. 3., 829 p., pgs. 195-197.

Am. Soc. Civil Engrs. (1990). Evapotranspiration and Irrigation Water Requirements, Am. Soc. Civil Engrs. Manuals and Reports on Engineering Practice, Publ. No. 70., 332 p., pgs. 25-35,38,132-140,168,169,174,175.

Hill, Dr. R.W., Johns, E.L., and Frevert, D.K., (1983). "Comparison of Equations Used For Estimating Crop Evapotranspiration With Field Research." Dep. Rep., U.S. Bureau of Reclamation. Denver, Co. pgs. 2-6, 2-9 thru 2-14.

McVay, M., U.S. Bureau of Reclamation, Boise, Idaho, (1992). "AgriMet - An Automatic Agricultural Weather Monitoring System", written for presentation at the American Society of Agricultural Engineers, Pacific Northwest Section, 47th Annual Meeting, Bozeman, Montana, September 16-18, 1992.

Powers, A., U.S. Bureau of Reclamation, Boise, Idaho, (1992). "Crop Water Use Modeling Through Satellite Telemetry", written for presentation at the American Society of Agricultural Engineers International Summer Meeting in Charlotte, North Carolina, June 21-24, 1992.